AstraZeneca

Covariate Adjustment for nAb titers

Seth Seegobin

AstraZeneca

Dec 1st 2022

Hypothetical Immunobridging Trial

Reference Vaccination

1st

2nd

Test Vaccine

Reference Group of Subjects

- Study population in whom clinical endpoint efficacy was demonstrated
- Authorized/approved vaccine formulation, dose level, and regimen

Test Group of Subjects

- New study population, vaccine formulation, dose level, or regimen
- (ideally, limit the number of variables that differ from the reference group)

Immune response marker(s)

Immune response marker(s)

Comparison via statistical hypothesis testing

(non-inferiority or superiority of test group to reference group)

Immunobridging Success Criteria Examples

- Emergency use authorization of COVID-19 Vaccine for use in age group different to that studied in the main efficacy trial (i.e. paediatric age groups)
  - Effectiveness inferred by immunobridging vs. reference group using
    - neutralizing antibody GMTs (1.5-fold margin)
    - seroresponse rates (10% margin)
- Emergency use authorization of booster doses COVID-19 Vaccines
  - Effectiveness inferred by immunobridging vs. respective primary series reference group, using
    - neutralizing antibody GMTs (1.5-fold margin) and
    - seroresponse rates (10% margin)

Hypothetical Immunobridging Trial

Reference Vaccination

1st

2nd

Test Vaccine

Reference Group of Subjects

- Study population in whom clinical endpoint efficacy was demonstrated
- Authorized/approved vaccine formulation, dose level, and regimen

Test Group of Subjects

- New study population, vaccine formulation, dose level, or regimen
- (ideally, limit the number of variables that differ from the reference group)

Immune response marker(s)

Immune response marker(s)

Comparison via statistical hypothesis testing

(non-inferiority or superiority of test group to reference group)

UK COVID-19 vaccines delivery plan

Phase 1 (Earlier vaccinations)	• Over-50s • care home residents • healthcare workers • people required to shield and others with certain health conditions
Phase 2 (Later vaccinations)	All those aged 40-49 years ⬇ All those aged 30-39 years ⬇ All those aged 18-29 years

Hypothetical Immunobridging Trial

Reference Group of Participants

- Younger
- Fewer comorbidities
- Gender differences in health seeking behaviour

Boost group of Participants

- Older
- Higher incidence of comorbidities
- Gender differences in health seeking behaviour

Randomisation cannot protect against baseline imbalance in prognostic participant characteristics

Primary Endpoint: Non-inferiority - GMT



$$H_0: \frac{GMT_{\text{test}}}{GMT_{\text{reference}}} \leq 2/3$$

$$H_A: \frac{GMT_{\text{test}}}{GMT_{\text{reference}}} > 2/3$$

Or equivalently

$$H_0: \log(GMT_{\text{test}}) - \log(GMT_{\text{reference}}) \leq \log(\frac{2}{3})$$

$$H_A: \log(GMT_{\text{test}}) - \log(GMT_{\text{reference}}) > \log(\frac{2}{3})$$

Potential challenge with baseline imbalance

- Within the UK, >65 were the first people vaccinated
  - Older participants are more likely to have co-morbidities

- Enrolment to the reference vaccination arm is heavily skewed towards younger persons with fewer co-morbidities as compared with the test arm
- As older age and co-morbidities are independently associated with lower immune response ignoring the baseline imbalance will make non-inferiority a higher hurdle (‘deck is stacked’ in favour of reference vaccination)

Potential options mitigate imbalance



Potential options mitigate imabalnce



Analyses of Covariance

- Comparing k means *adjusting* for 1 or more other variables (covariates)
- Compute adjusted (or least square) means.
- Sometimes called ANCOVA

Uses

- ➢To adjust for an imbalance among treatments in a baseline factor.
- ➢In observational studies controlling for a confounding factor
- ➢To throw light on the nature of treatment effects in a randomized study
- ➢Improve precision of estimated differences among treatments

Simulation

```
data simulation;
call streaminit(1979);
do TRT=0 to 1;
do i=1 to 350;
if TRT = 0 then do;
sex=Rand('BINOMIAL', 0.6, 1);
comorb=Rand('BINOMIAL', 0.1, 1);
age=Rand("normal",25,10);
end;
else if TRT=1 then do;
sex=Rand('BINOMIAL', 0.5, 1);
comorb=Rand('BINOMIAL', 0.65, 1);
age=Rand("normal",60,10);
end;
e=Rand("normal",0,1.5);
if age ge 65 then agegrp=1;
else if age lt 65 then agegrp=0;
logaval=4 + 0.5*trt + 0.25*sex + -0.5*comorb + -0.02*age + e;
output;
end;
end;
run;
```

	age	
	Mean	Std
TRT		
Reference	25.30	10.37
Test	59.53	10.00

	comorb	
	No	Yes
	PctN	PctN
TRT		
Reference	72	11
Test	28	89

	sex	
	Female	Male
	PctN	PctN
TRT		
Reference	43	55
Test	57	45

Geometric Mean Titre by Covariate



Analysis ignoring covariates (not advisable!)

```
proc mixed data=simulation;
           class trt(ref='0') AGEGRP(ref='0') sex(ref='1') comorb(ref='0');
           model logaval = trt /solution e3 residual outp=pred;
           lsmeans trt / e cl diff;
           lsmestimate TRT 1 -1 / testvalue=-0.176091259 upper CL alpha=0.025;
run;
```

Solution for Fixed Effects

Effect	TRT	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept		3.5867	0.08011	698	44.77	<.0001
TRT	1	-0.4304	0.1133	698	-3.80	0.0002
TRT	0	0	.	.	.	.

Test vaccination is associated with lower immune response

Least Squares Means

Effect	TRT	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	3.1563	0.08011	698	39.40	<.0001	0.05	2.9990	3.3136
TRT	0	3.5867	0.08011	698	44.77	<.0001	0.05	3.4294	3.7440

LSMEANS TRT;

Differences of Least Squares Means

Effect	TRT	_TRT	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	0	-0.4304	0.1133	698	-3.80	0.0002	0.05	-0.6529	-0.2080

Least Squares Means Estimate

Effect	Label	Estimate	Standard Error	DF	Test Value	t Value	Tails	Pr > t	Alpha	Lower	Upper
TRT	Row 1	-0.4304	0.1133	698	-0.176	-2.24	Upper	0.9875	0.025	-0.6529	Infty

LSMESTIMATE TRT;

Log10(2/3)

Fail non-inferiority

Analysis including covariates

```
proc mixed data=simulation;
        ODS output LSMEANS=LSMEANS SolutionF=SolutionF SOLUTIONR=SOLUTIONR;
        class trt(ref='0') sex(ref='1') comorb(ref='0');
        model logaval = trt age sex COMORB /solution e3 residual outp=pred;
        lsmeans trt / e cl diff;
        lsmestimate TRT 1 -1 0 0 0 0  / testvalue=-0.176091259 upper CL alpha=0.025;
run;
```

Solution for Fixed Effects

Effect	TRT	sex	comorb	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept				4.1758	0.1671	695	24.99	<.0001
TRT	1			0.5405	0.2389	695	2.26	0.0240
TRT	0			0	.	.	.	.
age				-0.02036	0.005508	695	-3.70	0.0002
sex		0		-0.1039	0.1139	695	-0.91	0.3620
sex		1		0	.	.	.	.
comorb			1	-0.4697	0.1439	695	-3.27	0.0011
comorb			0	0	.	.	.	.

Least Squares Means

Effect	TRT	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	3.5658	0.1256	695	28.39	<.0001	0.05	3.3192	3.8125
TRT	0	3.0253	0.1415	695	21.38	<.0001	0.05	2.7476	3.3031

LSMEANS TRT;

Differences of Least Squares Means

Effect	TRT	_TRT	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	0	0.5405	0.2389	695	2.26	0.0240	0.05	0.07144	1.0095

Test vaccination is associated with higher immune response

Least Squares Means Estimate

Effect	Label	Estimate	Standard Error	DF	Test Value	t Value	Tails	Pr > t	Alpha	Lower	Upper
TRT	Row 1	0.5405	0.2389	695	-0.176	3.00	Upper	0.0014	0.025	0.07144	Infty

LSMESTIMATE TRT;

Log10(2/3)

Establish non-inferiority

Merge the model estimates to the subject level data

Merge on the estimates to prediction dataset

```
PROC SQL;
          create table predestimate as
          select *,

          (select Estimate from solutionf where effect="Intercept") as E_intercept,
          (select Estimate from solutionf where effect="age") as E_age,
          (select Estimate from solutionf where effect="TRT" and Estimate^=0) as E_TRT,
          (select Estimate from solutionf where effect="sex" and Estimate^=0) as E_SEX,
          (select Estimate from solutionf where effect="comorb" and Estimate^=0) as   E_comorb
          from pred
          ;
quit;
```

Solution for Fixed Effects

Effect	TRT	sex	comorb	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept				4.1758	0.1671	695	24.99	<.0001
TRT	1			0.5405	0.2389	695	2.26	0.0240
TRT	0			0	.	.	.	.
age				-0.02036	0.005508	695	-3.70	0.0002
sex		0		-0.1039	0.1139	695	-0.91	0.3620
sex		1		0	.	.	.	.
comorb			1	-0.4697	0.1439	695	-3.27	0.0011
comorb			0	0	.	.	.	.

Obs	TRT	sex	comorb	age	logaval	Pred	E_intercept	E_age	E_TRT	E_SEX	E_comorb
1	0	1	0	18.4301	4.65936	3.80050	4.07187	-0.020361	0.54048	0.10390	-0.46974
2	0	0	0	36.5042	2.30926	3.32860	4.07187	-0.020361	0.54048	0.10390	-0.46974
3	0	0	0	18.1405	5.35951	3.70251	4.07187	-0.020361	0.54048	0.10390	-0.46974
4	0	1	0	23.3727	5.59988	3.69987	4.07187	-0.020361	0.54048	0.10390	-0.46974
5	0	1	0	15.1119	5.49800	3.86807	4.07187	-0.020361	0.54048	0.10390	-0.46974
6	0	1	0	24.7127	2.13682	3.67258	4.07187	-0.020361	0.54048	0.10390	-0.46974
7	0	1	0	11.5972	2.89502	3.93963	4.07187	-0.020361	0.54048	0.10390	-0.46974
8	0	1	0	21.0053	2.97697	3.74807	4.07187	-0.020361	0.54048	0.10390	-0.46974
9	0	1	0	17.9685	2.17933	3.80990	4.07187	-0.020361	0.54048	0.10390	-0.46974
10	0	1	0	7.9279	5.60223	4.01434	4.07187	-0.020361	0.54048	0.10390	-0.46974

Adjusted values

```
PROC SQL;
          create table recreate(keep=logaval mraw pred mpred adjy) as
          select *, mean(age) as mage,
          PRED + resid as mraw,
          sum(
          E_intercept,
          E_TRT*(TRT),
          E_age*(AGE),
          E_sex*(sex),
          E_comorb*(comorb)
)
          as MPRED,

     logaval - sum(
          E_age*(AGE-calculated mage),
          E_sex*(Sex-0.5),
          E_comorb*(comorb-0.5)
)
          as ADJY
          from predestimate
          ;
quit;
```

Predicted value from taking the estimate × subject specific covariate values

$$\overbrace{\hat{Y}_i}^{pred} = X\hat{\beta}$$

$$\overbrace{\tilde{Y}_i}^{adj} = \overbrace{Y_i}^{obs} - [(X - \bar{X})\hat{\beta}]$$

Adjusted values from taking observed values - subject specific covariate values minus the mean* covariate value × estimate

*mean of the range not mean of the value in the dataset

Predicted vs Adjusted Values

Uses a model to make the mean of X the same for all groups

What would the means have been if all groups had the same value of X (mean of X)?

```
Data recreate1;
        retain logaval Mraw PRED MPRED ADJY;
        set recreate;
        keep logaval Mraw PRED MPRED ADJY;
run;

Proc print data=recreate1 (obs=10);
run;
```

Obs	logaval	Mraw	PRED	MPRED	ADJY
1	4.65936	4.65936	3.80050	3.80050	3.88420
2	2.30926	2.30926	3.32860	3.32860	2.00601
3	5.35951	5.35951	3.70251	3.70251	4.68235
4	5.59988	5.59988	3.69987	3.69987	4.92536
5	5.49800	5.49800	3.86807	3.86807	4.65528
6	2.13682	2.13682	3.67258	3.67258	1.48958
7	2.89502	2.89502	3.93963	3.93963	1.98074
8	2.97697	2.97697	3.74807	3.74807	2.25425
9	2.17933	2.17933	3.80990	3.80990	1.39477
10	5.60223	5.60223	4.01434	4.01434	4.61324

LSMEANS vs Adjusted Values

Least Squares Means									
Effect	TRT	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	3.5658	0.1256	695	28.39	<.0001	0.05	3.3192	3.8125
TRT	0	3.0253	0.1415	695	21.38	<.0001	0.05	2.7476	3.3031

Differences of Least Squares Means										
Effect	TRT	_TRT	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	0	0.5405	0.2389	695	2.26	0.0240	0.05	0.07144	1.0095

```
PROC means data=recreate n mean std;
ods output summary=sum_rand;
class TRT;
var logaval Mraw PRED MPRED ADJY;
run;
```

The MEANS Procedure

TRT	N Obs	Variable	Label	N	Mean
0	350	logaval		350	3.5867353
		mraw		350	3.5867353
		Pred	Predicted	350	3.5867353
		MPRED		350	3.5867353
		ADJY		350	3.0253471
1	350	logaval		350	3.1563166
		mraw		350	3.1563166
		Pred	Predicted	350	3.1563166
		MPRED		350	3.1563166
		ADJY		350	3.5658242

Adjusted vs Observed values



Adjusted Means Computation Observations

$$\overbrace{\tilde{Y}_i}^{\text{adj}} = \overbrace{\mathrm{Y_i}}^{\text{obs}} - [(\mathrm{X} - \overline{\mathrm{X}})\hat{\beta}]$$

- If $\beta = 0$ then the adjusted value equals observed (unadjusted) value
- If the value of X is same for all participants, then adjusted value equals observed value (since $\bar{X} = X$)
- What would happen if we had used a value other than $\overline{\mathrm{X}}$?

Adjust to a specific covariate value

```
PROC SQL;
          create table recreate(keep=logaval mraw pred mpred adjy) as
          select *, mean(age) as mage,
          PRED + resid as mraw,
          sum(
          E_intercept,
          E_TRT*(TRT),
          E_age*(AGE),
          E_sex*(sex),
          E_comorb*(comorb)
)
          as MPRED,

     logaval - sum(
          E_age*(AGE-70),
          E_sex*(Sex-0.5),
          E_comorb*(comorb-0.5)
)
          as ADJY
          from predestimate
          ;
quit;
```

The MEANS Procedure (Age = 42.4)

TRT	N Obs	Variable	Label	N	Mean
0	350	logaval		350	3.5867353
		mraw		350	3.5867353
		Pred	Predicted	350	3.5867353
		MPRED		350	3.5867353
		ADJY		350	3.0253471
1	350	logaval		350	3.1563166
		mraw		350	3.1563166
		Pred	Predicted	350	3.1563166
		MPRED		350	3.1563166
		ADJY		350	3.5658242

The MEANS Procedure (Age = 70)

TRT	N Obs	Variable	Label	N	Mean
0	350	logaval		350	3.5867353
		mraw		350	3.5867353
		Pred	Predicted	350	3.5867353
		MPRED		350	3.5867353
		ADJY		350	2.4636542
1	350	logaval		350	3.1563166
		mraw		350	3.1563166
		Pred	Predicted	350	3.1563166
		MPRED		350	3.1563166
		ADJY		350	3.0041313

*mean of the range not mean of the value in the dataset

Adjust to a specific covariate value





Adjust to a specific covariate value

```
proc mixed data=simulation;
        class trt(ref='0') sex(ref='0') comorb(ref='0');
        model logaval = trt age sex COMORB /solution e3 residual outp=predold;
        lsmeans trt / at age=70 e cl diff;
        lsmestimate TRT 1 -1  / at AGE=70 testvalue=&testval. upper CL alpha=0.025;
run;
```

Least Squares Means										
Effect	TRT	age	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	70.00	3.0041	0.09885	695	30.39	<.0001	0.05	2.8100	3.1982
TRT	0	70.00	2.4637	0.2707	695	9.10	<.0001	0.05	1.9322	2.9951

Differences of Least Squares Means											
Effect	TRT	_TRT	age	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
TRT	1	0	70.00	0.5405	0.2389	695	2.26	0.0240	0.05	0.07144	1.0095

Least Squares Means Estimate											
Effect	Label	age	Estimate	Standard Error	DF	Test Value	t Value	Tails	Pr > t	Alpha	Lower
TRT	Row 1	70.00	0.5405	0.2389	695	-0.176	3.00	Upper	0.0014	0.025	0.07144

Because the covariate adjustment is identical between arms it does not change the estimate of the treatment difference

Simpler Method

$$LSMEAN = L\hat{\beta}$$

$$StdErr = L\left(X^T\hat{V}^{-1}X\right)^{-1}L^T$$

L matrix

```
lsmeans trt / e;
```

Coefficients for TRT Least Squares Means					
Effect	TRT	sex	comorb	Row1	Row2
Intercept				1	1
TRT	1			1	
TRT	0				1
age				42.414	42.414
sex		1		0.5	0.5
sex		0		0.5	0.5
comorb			1	0.5	0.5
comorb			0	0.5	0.5

L

$$\begin{matrix} Intercept & TRT_1 & TRT_0 & Age & Sex_1 & Sex_0 & Comorb_1 & Comorb_0 \end{matrix}$$

$$\begin{bmatrix} 1 & 1 & 0 & 42.414 & 0.5 & 0.5 & 0.5 & 0.5 \\ 1 & 0 & 1 & 42.414 & 0.5 & 0.5 & 0.5 & 0.5 \end{bmatrix}$$

$\hat{\beta}$ vector

Solution for Fixed Effects								
Effect	TRT	sex	comorb	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept				4.1758	0.1671	695	24.99	<.0001
TRT	1			0.5405	0.2389	695	2.26	0.0240
TRT	0			0	.	.	.	.
age				-0.02036	0.005508	695	-3.70	0.0002
sex		0		-0.1039	0.1139	695	-0.91	0.3620
sex		1		0	.	.	.	.
comorb			1	-0.4697	0.1439	695	-3.27	0.0011
comorb			0	0	.	.	.	.

$$\begin{matrix} Intercept \\ TRT_1 \\ TRT_0 \\ Age \\ Sex_1 \\ Sex_0 \\ Comorb_1 \\ Comorb_0 \end{matrix} \quad \begin{bmatrix} 4.1758 \\ 0.5405 \\ 0 \\ -0.02036 \\ -0.1039 \\ 0 \\ -0.4697 \\ 0 \end{bmatrix}$$

LSMeans

$$\begin{bmatrix} 1 & 1 & 0 & 42.414 & 0.5 & 0.5 & 0.5 & 0.5 \\ 1 & 0 & 1 & 42.414 & 0.5 & 0.5 & 0.5 & 0.5 \end{bmatrix} \begin{bmatrix} 4.1758 \\ 0.5405 \\ 0 \\ -0.02036 \\ -0.1039 \\ 0 \\ -0.4697 \\ 0 \end{bmatrix} = \begin{bmatrix} 3.5658 \\ 3.0253 \end{bmatrix}$$

Least Squares Means											
Effect	TRT	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	Cov1	Cov2
TRT	1	3.5658	0.1256	695	28.39	<.0001	0.05	3.3192	3.8125	0.01578	-0.01064
TRT	0	3.0253	0.1415	695	21.38	<.0001	0.05	2.7476	3.3031	-0.01064	0.02002

Conclusions

- Covariate adjustment provides a more accurate estimate of treatment effect in the presence of baseline imbalance with respect to response prognostic participant characteristics

- LSMEANS estimate the treatment effect assuming that covariate values are equal between treatment arms (default is the mean of X however any covariate values can be used)

Acknowledgments

I am indebted to the following individuals for their attempts to broaden my knowledge of covariate adjustment within Immunobridging trials:

- Kathryn Shoemaker
- Brett Jepson
- Dongmei Lan
- Sam Matthews
- Daniel Meddings

Thank you.